

New Developments in FeynCalc 9.0

Vladyslav Shtabovenko^{a,*}, Rolf Mertig^{b,**}, Frederik Orellana^c

^a*Technische Universität München, Physik-Department T30f, James-Franck-Str. 1, 85747 Garching, Germany*

^b*GluonVision GmbH, Bötzowstr. 10, 10407 Berlin, Germany*

^c*Technical University of Denmark, Anker Engelsevej 1, Building 101A, 2800 Kgs. Lyngby, Denmark*

Abstract

In this note we report on the new version of FEYNCALC, a MATHEMATICA package for symbolic semi-automatic evaluation of Feynman diagrams and algebraic expressions in quantum field theory. The main features of version 9.0 are: improved tensor reduction and partial fractioning of loop integrals, new functions for using FEYNCALC together with tools for reduction of scalar loop integrals using integration-by-parts (IBP) identities, better interface to FEYNARTS and support for $SU(N)$ generators with explicit fundamental indices.

Keywords: High energy physics; Feynman diagrams; Loop integrals; Dimensional regularization; Dirac algebra; Color algebra; Tensor reduction;

1. Introduction

In the last decades the importance of computer tools for higher order perturbative calculations in quantum field theory (QFT) has increased tremendously. Indeed, some recent achievements [1, 2, 3] in this field would have been hardly possible to complete within a reasonable time frame, if such projects were to be carried out only by pen and paper. The question that most QFT practitioners pose themselves today is not whether to use software tools or not, but rather which combination of tools will be the most efficient for the endeavored project. It is clear, that, in principle, there can be no

**E-mail address:* v.shtabovenko@tum.de

***E-mail address:* rolfin@gluonvision.com

universal package to cover any demand of any particle theorist. Instead, specific programs that provide different level of automation should be used for specific tasks. One of such specific tools is FEYNALC [4], that recently was released in the version 9.0.

FEYNALC is a MATHEMATICA package for algebraic calculations in QFT and semi-automatic evaluation of Feynman diagrams. The very first public version of FEYNALC, developed by Rolf Mertig with guidance from Ansgar Denner and Manfred Böhm, appeared in 1991. The main developments and improvements between 1991 and 1996 were triggered by the work of Mertig in electroweak theory [5, 6, 7, 8] and perturbative QCD [9]. Between 1997 and 2000, important contributions to the project came from Frederik Orellana, who, besides work on the general code, contributed the sub-package PHI for using FEYNALC in Chiral Perturbation Theory (χ PT) [10] and interfacing to FEYNARTS 3 [11]. From 2001 until 2014, with both developers out of theoretical physics, the development of FEYNALC was mostly constrained to bug fixing and providing support through the mailing list ¹, although some interesting projects with external collaborators still were conducted [12]. In 2014, the developer team was joined by Vladyslav Shtabovenko, who started to work on rewriting some parts of the existing code and implementing new features. In the same year the source code repository of FEYNALC was moved to GITHUB ², where the master branch of the repository represents the current development snapshot of the package. Not only the stable releases, but also the development version of FEYNALC can be anonymously downloaded by everyone at any time free of charge. The code is licensed under the Lesser General Public License (LGPL) version 3. To minimize the number of new bugs and regressions, an extensive unit testing framework ³ with over 3000 tests was introduced.

This note is organized in the following way. Section 2 compares FEYNALC to other packages for automatic evaluation of 1-loop Feynman diagrams and discusses setups, in which FEYNALC can be particularly useful. Section 3 provides an overview of interesting new features and improvements in FEYNALC 9.0. Section 4 gives an example of using FEYNALC to determine matching coefficients in NRQCD [18], a non-relativistic effective field

¹<http://www.feyncalc.org/forum>

²<https://github.com/FeynCalc>

³<https://github.com/FeynCalc/feyncalc/tree/master/Tests>

theory (EFT) for heavy quarkonia. Finally, we summarize and draw our conclusions in Section 5.

2. Comparison to similar tools

In view of the existence of several publicly available symbolic packages (FORMCALC [13], GOSAM [14], GRACE [15], DIANA [16]) that offer almost fully automatic evaluation of Feynman diagrams at 1-loop from Lagrangian till the cross-section, it appears necessary to explain how FEYNALC differs from such tools and why it is useful.

FEYNALC by itself does not provide a fully automatic way of computing cross sections or decay rates. Indeed, FEYNALC cannot generate Feynman diagrams and has no built-in capabilities for the numerical evaluation of master integrals and for the phase space integration. Therefore, these two important steps should be done using other tools.

Second, FEYNALC normally performs all the algebraic manipulations using MATHEMATICA. This leads to a slower performance when compared to tools that rely e.g. on FORM [17] for the symbolics. Despite some possibilities [12] to link FEYNALC with FORM, one should keep in mind that FEYNALC is not very well suited for evaluating hundreds, thousands or even millions of Feynman diagrams.

Finally, FEYNALC doesn't impose any particular ordering in which different parts (Dirac matrices, $SU(N)$ matrices, loop integrals etc.) of the amplitudes are supposed to be computed. It is always up to the user to decide what is the most useful way to carry out the calculation. This particular feature makes FEYNALC very different from tools that attempt to automatize all the steps of the evaluation process. Such tools usually stick to a particular workflow which roughly consists of the following steps:

1. The user specifies the process that needs to be computed.
2. If the given process is available in the standard configuration, load the corresponding model (e.g. Standard model (SM)). Otherwise the user must create a new model that contains this process.
3. Using the loaded model, generate relevant Feynman diagrams for the given process.
4. Evaluate the amplitudes by performing all the necessary algebraic simplifications.

5. Square the amplitude and sum/average over the spins of the involved particles.
6. Integrate over the phase space.

In this list, already the second step might turn out to be problematic. The list of built-in models usually includes SM and some popular (e.g. SUSY inspired) extensions, while more exotic theories require custom model files to be added by the user. If the Lagrangian of such a theory looks very different from \mathcal{L}_{SM} (e.g. in EFTs that are not strictly renormalizable (with an arbitrary number of legs in vertices) like χPT or even not manifestly Lorentz covariant like non-relativistic QCD (NRQCD) [18] or potential non-relativistic QCD (pNRQCD) [19]), then its implementation becomes a formidable task. On the other hand, even if the model can be implemented with a limited amount of effort, it still might cost more time than just writing down the amplitudes by hand and then manually entering them into the program. Although it is of course possible to make fully automatic tools accept such amplitudes as input, this is usually much less straight-forward than the official way of just specifying the process, launching the diagram generator and letting the automatics do the rest.

FEYNALC avoids such difficulties by accepting any kind of input that consists of valid FEYNALC objects. Hence, one can enter e.g. standalone Dirac traces, Lorentz vectors or loop integrals and then manipulate them with suitable FEYNALC functions. In this sense FEYNALC can be used much like a “calculator” for QFT expressions.

For manual input of Feynman diagrams FEYNALC contains some functions (`FeynRule`, `FunctionalD`, `CovariantD`, `QuantumField` etc.) for deriving Feynman rules from Lagrangians that are manifestly Lorentz covariant. At the same time it is of course also possible to evaluate Feynman diagrams that were generated automatically (e.g. by FEYNARTS), so that the user always can choose the most efficient strategy to get the calculation done.

Steps 4 and 5 usually imply that the user is not supposed to interfere too much with the evaluation process. Instead, one should rely on the available options to influence the outcome of the calculation. For example, when an automatic tool handles the Dirac algebra, it would normally try to simplify everything it can. While in general, this approach is perfectly fine, sometimes one would like to simplify only some of the Dirac structures, leaving the others (e.g. all the traces involving an odd number of γ^5) untouched. In principle, provided that the particular tool is open source, one can always modify its

code accordingly to obtain the desired output. Depending on the complexity of the code and the amount of documentation, this might, however, take some time and even introduce new bugs.

With FEYN CALC, the same result can be achieved in a more simple way, as one always has full access to all kind of intermediate expressions. For this purpose FEYN CALC also provides various helper functions (e.g. `Collect2`, `Expand2`, `Factor2`, `Isolate`, `ExpandScalarProduct`, `DiracGammaExpand`, `MomentumCombine`, `FCLoopSplit`, `FCLoopIsolate`, `FCLoopExtract`) that can be used to expand, sort, abbreviate and collect the given expressions with respect to particular structures.

Thus we see that FEYN CALC should not be regarded as a direct competitor to highly automatized packages like e.g. FORM CALC, because it neither provides routines for numerical evaluation nor offers a fully automatic workflow to evaluate a scattering process.

For studies that can be carried out using an automatic tool from the beginning to the end, it obviously would not be very efficient to stick to FEYN CALC. While one certainly can chain FEYN CALC with appropriate tools and libraries to obtain the same result, this would require more time and effort - which could be invested elsewhere.

The real niche that FEYN CALC fills are calculations that are too specific to be done in a fully automatic fashion but also too challenging to be done (only) by pen and paper, so that semi-automatic evaluation is very welcome.

One example for such problems is the determination of matching coefficients in EFTs. Matching coefficients are extracted by comparing suitable quantities (e.g. Green's functions) between the higher energy theory and its EFT at energies, where both theories should agree by construction. Then the quantity in the higher energy theory usually needs to be expanded in small scales and massaged into a form that resembles the same quantity in the lower energy theory, so that one can read off the values of the matching coefficients.

Such calculations are usually too special to be automatized in a full generality, but they can benefit a lot from functions provided by FEYN CALC. This is one of the reasons, why FEYN CALC enjoys certain popularity in the heavy quarkonium physics, where it is used to perform the matching between QCD and NRQCD for production [20, 21, 22] or decay [23, 24] of heavy quarkonia. Other studies where FEYN CALC was used involve such fields as Higgs [25, 26, 27] and top quark physics [28], phenomena in hadronic interactions [29, 30], dark matter [31, 32], neutrino physics [33, 34, 35] or gravity [36]. It

is worth noticing that FEYNCalc was also used at some stages of NNLO [37, 38] calculations. Indeed, FEYNCalc can be well employed for small or medium size multi-loop processes if one connects it to suitable tools for IBP-reduction (e.g. FIRE [39]) and numeric evaluation of multi-loop integrals (e.g. FIESTA [40] or SECDEC [41]).

Last but not least, FEYNCalc can be also quite useful for educational purposes. The possibility of easily getting hands-on experience with computing Feynman diagrams and exploring the different steps involved can be very helpful and motivating for students of quantum field theory.

3. New features in FeynCalc 9.0

3.1. Improved tensor decomposition

In the very early versions of FEYNCalc, tensor decomposition of 1-loop integrals (using Passarino-Veltman technique [42]) could be done only using the function `OneLoop`, where the maximal rank of the integrals was limited to 3 and the output was always written in terms of Passarino-Veltman coefficient functions. While working on [9], Rolf Mertig added to FEYNCalc 3.0 a tool (`Tdec`) for tensor decomposition of multi-loop integrals of arbitrary rank and multiplicity (for non-zero Gram determinants) and even included a database (`TIDL`) to load already computed decompositions, but only a very small amount of this functionality was turned into a user-friendly routine `TID` (1-loop only), while the rest remained to “lie idle” in the source code. `TID` was limited to 4-point functions of rank 4 and could not handle kinematic configurations with zero Gram determinants, so that for such cases one was forced to use `OneLoop`.

In FEYNCalc 9.0 `TID` was rewritten almost from scratch to allow for 1-loop tensor decompositions of any rank and multiplicity. At the beginning, the function computes Gram determinants for all the unique 1-loop integrals in the expression. If the determinant vanishes, the decomposition for that integral is done in terms of the Passarino-Veltman coefficient functions.

```
In[1]:= FCClearScalarProducts[];
        ScalarProduct[p1, p1] = 0;
        int = FCI[SPD[p2, q] FAD[{q, m0}, {q + p1, m1}]]
```

```
Out[1]:=  $\frac{p^2 \cdot q}{(q^2 - m0^2) \cdot ((p1 + q)^2 - m1^2)}$ 
```

In[2]:= TID[int, q]

Out[2]:= $i\pi^2(p_1 \cdot p_2)B_1(0, m_0^2, m_1^2)$

Otherwise, TID will output the result in terms of scalar 1-loop integrals.

In[1]:= FCClearScalarProducts[;
int = FCI[SPD[p2, q] FAD[{q, m0}, {q + p1, m1}]]

Out[1]:= $-\frac{(m_0^2 - m_1^2 + p_1^2)(p_1 \cdot p_2)}{2p_1^2(q^2 - m_0^2) \cdot ((q - p_1)^2 - m_1^2)} + \frac{p_1 \cdot p_2}{2p_1^2(q^2 - m_0^2)} - \frac{p_1 \cdot p_2}{2p_1^2(q^2 - m_1^2)}$

If needed, those scalar integrals can be converted to Passarino-Veltman scalar functions by using ToPaVe, which is also available since FEYN CALC 9.0.

In[2]:= TID[int, q] // ToPaVe[#, q] &

Out[2]:= $-\frac{i\pi^2(m_0^2 - m_1^2 + p_1^2)(p_1 \cdot p_2)B_0(p_1^2, m_0^2, m_1^2)}{2p_1^2} + \frac{i\pi^2 A_0(m_0^2)(p_1 \cdot p_2)}{2p_1^2} - \frac{i\pi^2 A_0(m_1^2)(p_1 \cdot p_2)}{2p_1^2}$

The decompositions in terms of scalar integrals tend to become very large already for 3-point functions, so to obtain more compact expressions it might be desirable to use the basis of Passarino-Veltman coefficient functions, even if there are no zero Gram determinants. This can be easily achieved via the option UsePaVeBasis.

In[1]:= int = FCI[FVD[q, mu] FVD[q, nu] FAD[{q, m0}, {q + p1, m1}, {q + p2, m2}]]

Out[1]:= $\frac{q^{\mu} q^{\nu}}{(q^2 - m_0^2) \cdot ((p_1 + q)^2 - m_1^2) \cdot ((p_2 + q)^2 - m_2^2)}$

In[2]:= TID[int/(I*Pi^2), q, UsePaVeBasis -> True]

**Out[2]:= $g^{\mu\nu} C_{00}(p_1^2, -2(p_1 \cdot p_2) + p_1^2 + p_2^2, p_2^2, m_0^2, m_1^2, m_2^2)$
 $+ p_1^{\mu} p_1^{\nu} C_{11}(p_1^2, -2(p_1 \cdot p_2) + p_1^2 + p_2^2, p_2^2, m_0^2, m_1^2, m_2^2)$
 $+ (p_2^{\mu} p_1^{\nu} + p_1^{\mu} p_2^{\nu}) C_{12}(p_1^2, -2(p_1 \cdot p_2) + p_1^2 + p_2^2, p_2^2, m_0^2, m_1^2, m_2^2)$
 $+ p_2^{\mu} p_2^{\nu} C_{22}(p_1^2, -2(p_1 \cdot p_2) + p_1^2 + p_2^2, p_2^2, m_0^2, m_1^2, m_2^2)$**

All the Passarino-Veltman functions are defined as in LOOPTOOLS [13] and explicit definitions are encoded for functions with up to 5 legs. For integrals with even higher multiplicities the coefficient functions (denoted as GenPaVe) simply include the dependence on the external momenta that can be used to convert them to the LOOPTOOLS or any other convention.

In[1]: = int = **FCI[FVD[q, mu] FVD[q, nu] FAD[{q, m0}, {q, m1}, {q, m2}, {q, m3}, {q + p4, m4}, {q + p5, m5}, {q + p6, m6}]]**

Out[1]: = $(q^{\mu} q^{\nu}) / (q^2 - m0^2) \cdot (q^2 - m1^2) \cdot ((p2 + q)^2 - m2^2) \cdot ((p3 + q)^2 - m3^2) \cdot ((p4 + q)^2 - m4^2) \cdot ((p5 + q)^2 - m5^2) \cdot ((p6 + q)^2 - m6^2)$

In[2]: = **TID[int/(I*Pi^2), q, UsePaVeBasis -> True]**

Out[2]: =

$$g^{\mu\nu} \text{GenPaVe} \left(\{0, 0\}, \begin{pmatrix} 0 & m0 \\ p1 & m1 \\ p2 & m2 \\ p3 & m3 \\ p4 & m4 \\ p5 & m5 \\ p6 & m6 \end{pmatrix} \right) + p1^{\mu} p1^{\nu} \text{GenPaVe} \left(\{1, 1\}, \begin{pmatrix} 0 & m0 \\ p1 & m1 \\ p2 & m2 \\ p3 & m3 \\ p4 & m4 \\ p5 & m5 \\ p6 & m6 \end{pmatrix} \right) + \dots$$

Here, $\text{GenPaVe}[\{1, 1\}, \{0, m0\}, \{\text{Momentum}[p1], m1\}, \dots, \{\text{Momentum}[p6], m6\}]$ stands for the coefficient function of $p_1^{\mu} p_1^{\nu}$ in the tensor decomposition of

$$\int d^D q \frac{q^{\mu} q^{\nu}}{[q^2 - m_0^2][(q - p_1)^2 - m_1^2][(q - p_2)^2 - m_2^2] \cdots [(q - p_6)^2 - m_6^2]}. \quad (1)$$

Since this kind of output is useful if one explicitly wants to obtain coefficient functions defined in a different way than in `LOOPTOOLS`, it can be activated also for functions with lower multiplicities by setting the option `GenPaVe` of `TID` to `True`.

In[1]: = int = **FCI[FVD[q, mu] FVD[q, nu] FAD[{q, m0}, {q + p1, m1}]]**

Out[1]: = $\frac{q^{\mu} q^{\nu}}{(q^2 - m0^2) \cdot ((p1 + q)^2 - m1^2)}$

In[2]: = **TID[int/(I*Pi^2), q, UsePaVeBasis -> True, GenPaVe -> True]**

Out[2]: =

$$g^{\mu\nu} \text{GenPaVe} \left(\{0, 0\}, \begin{pmatrix} 0 & m0 \\ p1 & m1 \end{pmatrix} \right) + p1^{\mu} p1^{\nu} \text{GenPaVe} \left(\{1, 1\}, \begin{pmatrix} 0 & m0 \\ p1 & m1 \end{pmatrix} \right)$$

One should also keep in mind that `FEYNCalc` cannot perform any further simplifications of `GenPaVe` functions, because internally they are not recognized as Passarino-Veltman integrals (`PaVe`).

It is well known that for a general multi-loop multi-scale integral, tensor decomposition doesn't allow to cancel all the scalar products containing loop momenta in the numerator, as it is the case at 1-loop. Nevertheless, this technique is widely used also in calculations beyond 1-loop, especially if one needs to deal with integrals that have loop momenta contracted to Dirac matrices or epsilon tensors or even loop momenta with free Lorentz indices. FEYN CALC uses a quite simple reduction algorithm (implemented in `Tdec`) that consists of decomposing the integral into all tensor structures allowed by the symmetries and using Gaussian elimination to obtain the coefficients of each tensor.

Since tensor decomposition of multi-loop integrals with FEYN CALC's function `Tdec` is not quite straight-forward and usually requires some additional MATHEMATICA code, in FEYN CALC 9.0 a new function `FCMultiLoopTID` was added, that makes multi-loop tensor reduction work out of the box.

```
In[1]:= int = FCI[FVD[q1, mu] FVD[q2, nu] FAD[q1, q2, {q1 - p1},
                {q2 - p1}, {q1 - q2}]]
```

```
Out[1]:=  $\frac{q1^{mu} q2^{nu}}{q1^2 \cdot q2^2 \cdot (q1-p1)^2 \cdot (q2-p2)^2 \cdot (q1-q2)^2}$ 
```

```
In[2]:= FCMultiLoopTID[int, {q1, q2}]
```

```
Out[2]:=  $\frac{Dp1^{mu} p1^{nu} - p1^2 g^{munu}}{4(D-1)q2^2 \cdot q1^2 \cdot (q2-p1)^2 \cdot (q1-q2)^2 \cdot (q1-p1)^2} - \frac{Dp1^{mu} p1^{nu} - p1^2 g^{munu}}{2(D-1)p1^4 q1^2 \cdot (q2-p1)^2 \cdot (q1-q2)^2}$ 
 $+ \frac{p1^2 g^{munu} - p1^{mu} p1^{nu}}{(D-1)p1^2 q2^2 \cdot q1^2 \cdot (q1-q2)^2 \cdot (q1-p1)^2} - \frac{p1^2 g^{munu} - p1^{mu} p1^{nu}}{2(D-1)p1^2 q2^2 \cdot q1^2 \cdot (q2-p1)^2 \cdot (q1-p1)^2}$ 
```

Unfortunately, the reduction breaks down when the corresponding Gram determinant vanishes. For such cases, in a future version it is planned to include a more useful algorithm.

3.2. New partial fractioning algorithm

Since the version 3, FEYN CALC includes `ScalarProductCancel` and `Apart2` that can be used to rewrite loop integrals in a simpler form. `ScalarProductCancel` essentially applies the well known identity [42]

$$q \cdot p = \frac{1}{2}[(q+p)^2 + m_2^2 - (q^2 + m_1^2) - p^2 - m_2^2 + m_1^2] \quad (2)$$

repeatedly, until all scalar products containing loop momenta that can be canceled in this way are eliminated. `Apart2` uses the trivial identity

$$\frac{1}{(q^2 - m_1^2)(q^2 - m_2^2)} = \frac{1}{m_1^2 - m_2^2} \left(\frac{1}{q^2 - m_1^2} - \frac{1}{q^2 - m_2^2} \right) \quad (3)$$

to simplify suitable denominators. In principle, these two functions implement some aspects of partial fractioning, i.e., the decomposition of a loop integral with linearly dependent propagators into a sum of integrals where each integral contains only linearly independent propagators. Notice that here we count scalar products that involve loop momenta as propagators with negative exponents. Unfortunately, there are plenty of examples where neither `ScalarProductCancel` nor `Apart2` can partial fraction an integral with linearly dependent propagators, e.g.

$$\int d^D q \frac{1}{q^2(q-p)^2(q+p)^2} = \frac{1}{p^2} \int d^D q \left(\frac{1}{q^2(q-p)^2} - \frac{1}{(q-p)^2(q+p)^2} \right) \quad (4)$$

A general partial fractioning algorithm that is suitable for multi-loop integrals including its MATHEMATICA implementation (`APART`⁴) was presented in [43]. The author has also shown how his code can be used together with `FEYNCALC` in order to decompose different loop integrals. For this the user is required to convert a loop integral in the `FEYNCALC` notation (with denominator encoded in `FeynAmpDenominator`) to a somewhat different form and to specify all the scalar products that contain loop momenta and appear in this loop integral. After the decomposition the resulting integrals need to be converted back into `FEYNCALC` notation.

In `FEYNCALC` 9.0 the algorithm from [43] was adopted and reimplemented to be the standard partial fractioning routine. As such, it is fully integrated with all other `FEYNCALC` functions and objects and doesn't require any explicit conversion of the input or output.

```
In[1]:= int = FAD[{q}, {q - p}, {q + p}]
```

```
Out[1]:=  $\frac{1}{q^2 \cdot (q-p)^2 \cdot (p+q)^2}$ 
```

```
In[2]:= ApartFF[int1, {q}]
```

⁴<https://github.com/F-Feng/APart>

$$\text{Out}[2]:= \frac{1}{p^2 q^2 \cdot (q-p)^2} - \frac{1}{p^2 q^2 \cdot (q-2p)^2}$$

The name of the corresponding function is `ApartFF` which stands for “Apart Feng Feng” and serves as an additional acknowledgement of the original author. One should also notice that while the original `APART` can be used for partial fractioning of quite general multivariate polynomials, the `FEYN CALC` version is limited only to polynomials that appear in Feynman diagrams as propagators and scalar products. Thus, it is much less general than `APART` but is also more convenient when used with `FEYN CALC`.

3.3. Tools for interfacing `FEYN CALC` with packages for IBP-reduction

In modern multi-loop calculations, reduction of scalar loop integrals via integration-by-parts (IBP) identities [44] is a regular step needed to arrive to a smaller set of master integrals.

Although `FEYN CALC` doesn’t include a general purpose tool for IBP reduction (the built-in `TARCER` [45] is suitable only for 2-loop self-energy type integrals), this omission can be compensated by using one of the publicly available IBP-packages (`FIRE` [39], `LITERED` [46], `REDUZE` [47], `AIR` [48]). However, one should keep in mind that such tools usually expect their input to contain only loop integrals with linearly independent propagators that form a basis. For example, the integral

$$\int d^D q_1 d^D q_2 d^D q_3 \frac{1}{[q_1^2 - m^2]^2 [(q_1 + q_3)^2 - m^2] (q_2 - q_3)^2 q_2^2} \quad (5)$$

cannot be processed by `FIRE` in this form because q_1^2 , q_2^2 , $(q_1 + q_3)^2$ and $(q_2 - q_3)^2$ alone do not form a basis.

```
In[1]:= << FIRE5'FIRE5'
Internal = {q1, q2, q3};
External = {};
Propagators = {q1^2 - m^2, (q1 + q3)^2 - m^2, (q2 - q3)^2, q2^2};
PrepareIBP[];
```

```
Out[1]:= FIRE, version 5.1
DatabaseUsage: 0
UsingFermat: False
Not enough propagators. Add irreducible nominators
```

If one includes also q_3^2 and $q_1 \cdot q_2$ with zero exponentials, then we have a proper basis and the reduction works as it should. Also the integral

$$\int d^D q_1 d^D q_2 \frac{(p \cdot q_1)^2 (p \cdot q_2)^2}{[q_1^2 - m^2][q_2^2 - m^2](q_1 - p)^2 (q_2 - p)^2 (q_1 - q_2)^2} \quad (6)$$

cannot be reduced right away, this time because its propagators are linearly dependent.

```
In[1]:= << FIRE5'FIRE5'
      Internal = {q1, q2};
      External = {};
      Propagators = {q1^2 - m^2, q2^2 - m^2, (q1 - p)^2, (q2 - p)^2, (q1 - q2)^2, p q1, p q2};
      PrepareIBP[];
```

```
Out[1]:= FIRE, version 5.1
      DatabaseUsage: 0
      UsingFermat: False
      Linearly dependant propagators. Perform reduction first
```

To detect such problems before the reduction actually fails, FEYN CALC 9.0 introduces two new special functions. When `FCLoopBasisIncompleteQ` is applied to a loop integral, it returns `True` if this integral doesn't contain enough irreducible propagators.

```
In[1]:= intP1 = FCI[FAD[{q1, m, 2}, {q1 + q3, m}, {q2 - q3}, q2]]
```

```
Out[1]:=  $\frac{1}{(q1^2 - m^2) \cdot (q1^2 - m^2) \cdot ((q1 + q3)^2 - m^2) \cdot (q2 - q3)^2 \cdot q2^2}$ 
```

```
In[2]:= FCLoopBasisIncompleteQ[intP1, {q1, q2, q3}]
```

```
Out[2]:= True
```

```
In[3]:= FCLoopBasisIncompleteQ[SPD[q3, q3] SPD[q1, q2] intP1, {q1, q2, q3}]
```

```
Out[3]:= False
```

An integral with linearly dependent propagators will be detected by `FCLoopBasisOverdeterminedQ`,

```
In[1]:= intP2 = FCI[SPD[p, q1]^2 SPD[p, q2]^2 FAD[{q1, m}, {q2, m}, q1 - p, q2 - p, q1 - q2]]
```

```
Out[1]:=  $\frac{(p \cdot q1)^2 (p \cdot q2)^2}{(q1^2 - m^2) \cdot (q2^2 - m^2) \cdot (q1 - p)^2 \cdot (q2 - p)^2 \cdot (q1 - q2)^2}$ 
```

```
In[2]:= FCLoopBasisOverdeterminedQ[intP2, {q1, q2, q3}]
```

```
Out[2]:= True
```

so that only an integral for which both functions return `False` can be reduced in a straight-forward way.

Of course, in a practical calculation where one knows what integral topologies are involved, such issues can be easily resolved. In particular, a clever choice of additional propagators that are needed to have a basis, can greatly simplify the reduction. On the other hand, depending on the size of the problem and the number of topologies involved, a less clever but fully automatic solution may also be useful.

For an integral with linearly dependent propagators we can use `ApartFF`, that is guaranteed to decompose it into integrals where all propagators are linearly independent.

In[1]:= `ApartFF`[intP2, {q1, q2}]

$$\begin{aligned} \text{Out[1]:= } & \frac{(m^2+p^2)^4}{16(q1^2-m^2).(q2^2-m^2).(q2-p)^2.(q1-q2)^2.(q1-p)^2} \\ & - \frac{(m^2+p^2)^3}{8(q1^2-m^2).(q2^2-m^2).(q1-q2)^2.(q1-p)^2} \\ & - \frac{(m^2+p^2)(p \cdot q1)}{8(q2^2-m^2).(q1-q2)^2.(q1-p)^2} + \dots \end{aligned}$$

For integrals with an incomplete basis of propagators one can use the new function `FCLoopBasisFindCompletion` that finds out which irreducible propagators (with zero exponents) are missing.

In[1]:= `FCLoopBasisFindCompletion`[intP1, {q1, q2, q3}]

$$\text{Out[1]:= } \left\{ \frac{1}{(q1^2-m^2).(q1^2-m^2).((q1+q3)^2-m^2).(q2-q3)^2.q2^2}, \{-(q1 \cdot q3) + q2 \cdot q3 + 2q3^2, q1 \cdot q2\} \right\}$$

With the suggested propagators the integral is guaranteed to have a complete basis, but the choice of the propagators themselves is usually not very clever. This is because in general FEYNCalc cannot guess the topology of the given integral without any additional input. It is planned to provide a possibility for specifying the topology, which would admittedly make `FCLoopBasisFindCompletion` much more useful than it is now.

Still, with `ApartFF` and `FCLoopBasisFindCompletion` it is now possible to automatically bring any scalar multi-loop integral in FEYNCalc notation to a form that can be directly (modulo notation conversion) forwarded to an IBP tool.

3.4. Advanced extraction of loop integrals

The idea to use FEYN CALC as a sort of switch board for different computational tools in a larger framework (see e.g. [49]) is further developed in version 9.0 by the introduction of new functions that can extract different loop integrals from the given expression.

One of them, `FCLoopSplit`, breaks the given expression into 4 pieces, which are

1. Terms that contain no loop integrals.
2. Terms that only contain scalar loop integrals without any loop momenta in the denominators, e.g.

$$\int d^D q \frac{1}{q^2 - m^2}. \quad (7)$$

3. Terms that contain scalar loop integrals with loop momenta dependent scalar products in the denominators, e.g.

$$\int d^D q \frac{(q \cdot p)}{q^2 (q - p)^2} \quad (8)$$

4. Terms that contain tensor loop integrals, e.g.

$$\int d^D q \frac{q^\mu q^\nu}{q^2 - m^2} \quad \text{or} \quad \int d^D q \frac{(\gamma \cdot q)}{q^2 (q - p)^2} \quad (9)$$

In[1]:= int = **FCI**[(**GSD**[q - p] + m).**GSD**[x] **FAD**[q, {q - p, m}] + (m^2 + **SPD**[q, q])
FAD[{q, m, 2}]];

Out[1]:= $\frac{(m + \gamma \cdot (q - p)) \cdot (\gamma \cdot x)}{q^2 \cdot ((q - p)^2 - m^2)} + \frac{m^2 + q^2}{(q^2 - m^2) \cdot (q^2 - m^2)}$

In[2]:= **FCLoopSplit**[int, {q}]

Out[2]:= $\left\{ 0, \frac{m \gamma \cdot x - (\gamma \cdot p) \cdot (\gamma \cdot x)}{q^2 \cdot ((q - p)^2 - m^2)} + \frac{m^2}{(q^2 - m^2) \cdot (q^2 - m^2)}, \frac{q^2}{(q^2 - m^2) \cdot (q^2 - m^2)}, \frac{(\gamma \cdot q) \cdot (\gamma \cdot x)}{q^2 \cdot ((q - p)^2 - m^2)} \right\}$

This splitting makes it easier to handle different types of loop integrals and to simplify them with FEYN CALC or other tools. For example, if one wants to perform tensor reduction of multi-loop integrals with FaRe [50] instead of `FCMultiLoopTID`, it can be done by applying `FCLoopSplit` to the given expression and working with the 4th element of the resulting list, while the other elements remain unchanged and can be later added to the final expression.

To handle a larger number of loop diagrams in an efficient way, **FCLoopSplit** alone is not sufficient. This is because same integrals may appear multiple times in different diagrams and ignoring this fact would make the evaluation more complex than it actually is. To avoid this kind of problems one should better first analyze the amplitude and extract all the unique integrals. Then each unique integral needs to be evaluated only once, no matter how often it appears in the full expression. In **FEYN CALC 9.0** this can be conveniently done with **FCLoopIsolate**. The function wraps loop integers with the given head, such that the list of unique integrals can be quickly created with **MATHEMATICA**'s **Cases** and **Union** or just **FEYN CALC**'s **Cases2**

```
In[1]:= int = FCI[GSD[q - p1].(GSD[q - p2] + M).GSD[p3] SPD[q, p2] FAD[q, q - p1, {q - p2, m}]];
```

```
Out[1]:=  $\frac{(p2 \cdot q)(\gamma \cdot (q - p1)) \cdot (M + \gamma \cdot (q - p2)) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)}$ 
```

```
In[2]:= res = FCLoopIsolate[int, {q}, Head -> loopInt]
```

```
Out[2]:= loopInt  $\left( \frac{p2 \cdot q}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right) ((\gamma \cdot p1) \cdot (\gamma \cdot p2) \cdot (\gamma \cdot p3) - M(\gamma \cdot p1) \cdot (\gamma \cdot p3)) +$ 
```

```
 $M$ loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right) -$  loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot p1) \cdot (\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right)$ 
```

```
 $-$  loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p2) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right) +$  loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right)$ 
```

```
In[3]:= Cases2[res, loopInt]
```

```
Out[3]:=  $\left\{ \text{loopInt} \left( \frac{p2 \cdot q}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right), \text{loopInt} \left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right), \right.$ 
```

```
 $\text{loopInt} \left( \frac{(p2 \cdot q)(\gamma \cdot p1) \cdot (\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right), \text{loopInt} \left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p2) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right),$ 
```

```
 $\left. \text{loopInt} \left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q - p1)^2 \cdot ((q - p2)^2 - m^2)} \right) \right\}$ 
```

The combined application of **FCLoopIsolate** and **FCLoopSplit** is provided by **FCLoopExtract**. This function returns a list of three entries. The first one contains the part of the expression which is free of loop integrals. The second entry consists of the remaining expression where every loop integral is wrapped with the given head. Finally, the last entry contains a list of all the unique loop integrals in the expression.

```
In[4]:= FCLoopExtract[int, {q}, loopInt ][[1]]
Out[4]:= 0
```

```
In[5]:= FCLoopExtract[int, {q}, loopInt ][[2]] ===
FCLoopIsolate[int, {q}, Head -> loopInt]
Out[5]:= True
```

```
In[6]:= FCLoopExtract[int, {q}, loopInt ][[3]] ===
Cases2[res, loopInt]
Out[6]:= True
```

Suppose that we want to evaluate these loop integrals using some custom function `loopEval` (in this example it is just a dummy function that computes the hash of each loop integral). All we need to do is to apply `FCLoopExtract` to the initial expression, map the list of the unique integrals to `loopEval`, create a substitution rule and apply this rule to our expression in order to get the final result.

```
In[7]:= {rest, loops, intsUnique} = FCLoopExtract[int, {q}, loopInt];
```

```
In[8]:= loopEval[x_] := ToString[Hash[x]];

```

```
In[9]:= solsList = loopEval /@ uniqueInts
```

```
Out[9]:= {2069116068,115167616,776830638,1878762839,1337833147}
```

```
In[10]:= repRule = MapThread[Rule[#1, #2] &, {intsUnique, solsList}]
```

```
Out[10]:= {loopInt  $\left( \frac{p^2 \cdot q}{q^2 \cdot (q-p1)^2 \cdot ((q-p2)^2 - m^2)} \right) \rightarrow 2069116068,$ 
```

```
loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q-p1)^2 \cdot ((q-p2)^2 - m^2)} \right) \rightarrow 115167616,$ 
```

```
loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot p1) \cdot (\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q-p1)^2 \cdot ((q-p2)^2 - m^2)} \right) \rightarrow 776830638,$ 
```

```
loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot p2) \cdot (\gamma \cdot p3)}{q^2 \cdot (q-p1)^2 \cdot ((q-p2)^2 - m^2)} \right) \rightarrow 1878762839,$ 
```

```
loopInt  $\left( \frac{(p2 \cdot q)(\gamma \cdot q) \cdot (\gamma \cdot q) \cdot (\gamma \cdot p3)}{q^2 \cdot (q-p1)^2 \cdot ((q-p2)^2 - m^2)} \right) \rightarrow 1337833147}$ 
```

```
In[11]:= res = rest + loops /. repRule
```

```
Out[11]:= 115167616M + 1337833147 - 1878762839 +
```

```
2069116068((\gamma \cdot p1) \cdot (\gamma \cdot p2) \cdot (\gamma \cdot p3) - M(\gamma \cdot p1) \cdot (\gamma \cdot p3)) - 776830638
```

With `FCLoopSplit`, `FCLoopIsolate` and `FCLoopEvaluate` it is now much

easier not only to manipulate loop integrals, but also to check which integrals actually appear in an expression. Unique loop integrals can be evaluated with tools outside of FEYN CALC and then substituted back by just a couple of lines of MATHEMATICA code.

3.5. Better interface to FEYNARTS

If FEYN CALC needs to be used with a Feynman diagram generator, then FEYNARTS is usually the most convenient choice. Initially the syntax of both packages was adjusted to make them fully compatible with each other. In fact, for the very first version of FEYNARTS [51], FEYN CALC was referred to as the standard tool to evaluate the generated amplitudes. As FEYNARTS was developed further, the full compatibility was lost, but even now, the output of FEYNARTS can be converted into valid FEYN CALC input with only little effort. A more severe problem in using this setup arises when FEYNARTS and FEYN CALC are loaded in the same MATHEMATICA session. Unfortunately, both packages contain objects with same names but different contexts, definitions and properties (e.g. `FourVector`, `DiracMatrix` or `FeynAmpDenominator`) such that it is not possible to use them together without risking inconsistencies. To avoid these issues FEYN CALC is able to automatically patch the source code of FEYNARTS by renaming all the conflicting symbols, such that e.g. `FourVector` becomes `FAFourVector` and no variable shadowing can occur. This patching mechanism was greatly improved in FEYN CALC 9.0 both in terms of user friendliness and compatibility to other MATHEMATICA packages. The patched copy of FEYNARTS now resides in the *FeynArts* directory inside the FEYN CALC installation. By default this directory is empty. The user is expected to manually download the latest FEYNARTS tarball from the official website⁵ and unpack its content to *FeynCalc/FeynArts*. When FEYN CALC is loaded via

```
$LoadFeynArts=True;  
<<FeynCalc'
```

it will automatically detect FEYNARTS installation and offer the user to patch it. This procedure is required only once and after that one can use FEYNARTS and FEYN CALC together without any problems.

⁵<http://www.feynarts.de>

After all the required diagrams have been generated and turned into amplitudes with FEYNARTS' function `CreateFeynAmp`, the output still needs to be converted into valid FEYNCALC input. In FEYNCALC 9.0 this is handled by the new function `FCFAConvert` that takes the output of `CreateFeynAmp` and generates proper FEYNCALC expressions based on the given options. With `IncomingMomenta`, `OutgoingMomenta` and `LoopMomenta` the user can specify how the corresponding momenta should be named. Otherwise they will be denoted as `InMom1`, `InMom2`, `...`, `OutMom1`, `OutMom2`, `...` and `LoopMom1`, `LoopMom2`, `...`. Polarization vectors of external massless bosons are by default not transverse, but can be made so if the momenta of the bosons are listed in `TransversePolarizationVectors`. The splitting of fermion-fermion-boson couplings into left and right handed chirality projectors (default in FEYNARTS) can be undone with the option `UndoChiralSplittings`. For example, the amplitude for the tree level process $\gamma^* u \rightarrow u g$ is obtained via

```
In[1]:= $LoadFeynArts = True;
        $FeynCalcStartupMessages = False;
        << FeynCalc';
        $FAVerbose = 0;

In[2]:= diags = InsertFields[ CreateTopologies[0, 2 -> 2], {F[3, {1}],
        V[1]} -> {V[5], F[3, {1}]}, InsertionLevel -> {Classes},
        Model -> "SMQCD"];

In[3]:= FCFAConvert[CreateFeynAmp[diags], IncomingMomenta -> {p1, kp},
        OutgoingMomenta -> {kg, p2}, UndoChiralSplittings -> True,
        TransversePolarizationVectors -> {kg}, DropSumOver -> True,
        List -> False] // Contract

Out[3]:= 
$$-\frac{2ELg_s T_{Col4Col1}^{Glu3}(\varphi(\overline{p2}, MU)) \cdot (\tilde{\gamma} \cdot \tilde{\epsilon}^*(kg)) \cdot (\tilde{\gamma} \cdot (\overline{kg} + \overline{p2}) + MU) \cdot (\tilde{\gamma} \cdot \tilde{\epsilon}(kp)) \cdot (\varphi(\overline{p1}, MU))}{3((-\overline{kg} - \overline{p2})^2 - MU^2)}$$


$$-\frac{2ELg_s T_{Col4Col1}^{Glu3}(\varphi(\overline{p2}, MU)) \cdot (\tilde{\gamma} \cdot \tilde{\epsilon}(kp)) \cdot (\tilde{\gamma} \cdot (\overline{p2} - \overline{kp}) + MU) \cdot (\tilde{\gamma} \cdot \tilde{\epsilon}^*(kg)) \cdot (\varphi(\overline{p1}, MU))}{3((\overline{kp} - \overline{p2})^2 - MU^2)}$$

```

3.6. Finer-grained expansions

To expand scalar products of Lorentz vectors FEYNCALC provides the function `ExpandScalarProduct`. The standard behavior of this command is to expand every scalar product in the expression.

```
In[1]:= exp = SPD[q1, p1 + p2] SPD[q2, p3 + p4] SPD[p5 + p6, p7 + p8]

Out[1]:= ((p1 + p2) · q1)((p3 + p4) · q2)((p5 + p6) · (p7 + p8))
```

In[2]:= ExpandScalarProduct[exp]

Out[2]:= $(p_1 \cdot q_1 + p_2 \cdot q_1)(p_3 \cdot q_2 + p_4 \cdot q_2)(p_5 \cdot p_7 + p_5 \cdot p_8 + p_6 \cdot p_7 + p_6 \cdot p_8)$

which might lead to an unnecessary increase of terms, if the user wants to expand only some particular scalar products. FEYN CALC 9.0 improves `ExpandScalarProduct` by introducing the option `Momentum` which allows to specify a list of momenta that need to be contained in a scalar product that will be expanded. All the other scalar products will remain untouched.

In[1]:= `exp = SPD[q1, p1 + p2] SPD[q2, p3 + p4] SPD[p5 + p6, p7 + p8]`

Out[1]:= $((p_1 + p_2) \cdot q_1)((p_3 + p_4) \cdot q_2)((p_5 + p_6) \cdot (p_7 + p_8))$

In[2]:= ExpandScalarProduct[exp, Momentum -> {q1}]

Out[2]:= $(p_1 \cdot q_1 + p_2 \cdot q_1)((p_3 + p_4) \cdot q_2)((p_5 + p_6) \cdot (p_7 + p_8))$

In[3]:= ExpandScalarProduct[exp, Momentum -> {q2}]

Out[2]:= $(p_3 \cdot q_2 + p_4 \cdot q_2)((p_1 + p_2) \cdot q_1)((p_5 + p_6) \cdot (p_7 + p_8))$

The same option is now present also in `DiracGammaExpand` that is used to expand Lorentz vectors contracted with Dirac matrices

In[1]:= `exp = GSD[q1 + p1 + p2].GSD[q2 + p3 + p4].GSD[p5 + p6 + p7 + p8]`

Out[1]:= $(\gamma \cdot (p_1 + p_2 + q_1)).(\gamma \cdot (p_3 + p_4 + q_2)).(\gamma \cdot (p_5 + p_6 + p_7 + p_8))$

In[2]:= DiracGammaExpand[exp]

Out[2]:= $(\gamma \cdot p_1 + \gamma \cdot p_2 + \gamma \cdot q_1).(\gamma \cdot p_3 + \gamma \cdot p_4 + \gamma \cdot q_2).(\gamma \cdot p_5 + \gamma \cdot p_6 + \gamma \cdot p_7 + \gamma \cdot p_8)$

In[3]:= DiracGammaExpand[exp, Momentum -> {q1}]

Out[3]:= $(\gamma \cdot p_1 + \gamma \cdot p_2 + \gamma \cdot q_1).(\gamma \cdot (p_3 + p_4 + q_2)).(\gamma \cdot (p_5 + p_6 + p_7 + p_8))$

In[4]:= DiracGammaExpand[exp, Momentum -> {q2}]

Out[4]:= $(\gamma \cdot (p_1 + p_2 + q_1)).(\gamma \cdot p_3 + \gamma \cdot p_4 + \gamma \cdot q_2).(\gamma \cdot (p_5 + p_6 + p_7 + p_8))$

3.7. $SU(N)$ generators with explicit fundamental indices

FEYN CALC denotes $SU(N)$ generators in the fundamental representation as `SUNT[a]` where `a` stands for the adjoint index. The fundamental indices are suppressed, so that a chain of `SUNT`-matrices is understood to have only

two free fundamental indices, e.g. `SUNT[a,b,c]` stands for $T_{ij}^a T_{jk}^b T_{kl}^c$ and it is not possible to express, say $T_{ij}^a T_{kl}^b$ with `SUNT` objects only.

Due to this limitation, evaluation of Feynman amplitudes with more than two free fundamental color indices (e.g. $q\bar{q} \rightarrow q\bar{q}$ scattering in QCD) was very inconvenient and usually required additional `MATHEMATICA` code to obtain the correct result. For this reason `FEYNCALC 9.0` introduces a new object `SUNTF[{a},i,j]` that stands for T_{ij}^a , an $SU(N)$ generator in the fundamental representation with explicit fundamental indices `i` and `j` and the adjoint index `a`. Hence expressions like $T_{ij}^a T_{kl}^b$ or $T_{ij}^a T_{jk}^b T_{lm}^c$ can be now conveniently expressed with `SUNTF[{a},i,j]*SUNTF[{b},k,l]` and `SUNTF[{a,b},i,k]*SUNTF[{c},l,m]` respectively. The new `SUNTF` objects are fully compatible with `SUNSimplify`, the standard routine for simplifying $SU(N)$ algebra.

```
In[1]:= exp1 = SUNTF[{a}, i, j] SUNTF[{b}, j, k] SUNTF[{c}, k, l]
```

```
Out[1]:= T_{ij}^a T_{jk}^b T_{kl}^c
```

```
In[2]:= SUNSimplify[exp1]
```

```
Out[2]:= (T^a T^b T^c)_{il}
```

```
In[3]:= exp2 = exp1 SUNFDelta[i, l]
```

```
Out[3]:= \delta_{il} T_{ij}^a T_{jk}^b T_{kl}^c
```

```
In[4]:= SUNSimplify[exp2]
```

```
Out[4]:= tr(T^c . T^a . T^b)
```

4. Using `FeynCalc` with non-relativistic EFTs

Up to now we silently assumed that all the amplitudes and expressions that we want to evaluate stem from a theory that is manifestly Lorentz covariant. This nice property of relativistic QFTs is often taken for granted, but one surely should not forget about EFTs that are used to describe non-relativistic systems, where the corresponding Lagrangians often do not exhibit manifest Lorentz covariance.

To our knowledge, there are no public tools for doing algebraic calculations in non-relativistic EFTs, where one has to explicitly distinguish be-

tween temporal and spatial components of Lorentz tensors. Naively, one might think that to do a calculation in such a theory using computer, one would need to write a large amount of additional code almost from scratch. However, with such a versatile tool like FEYN CALC, this estimate turns out to be too pessimistic. In the following we want to give a simple example of using FEYN CALC in a non-relativistic calculation, where only a comparably small amount of additional MATHEMATICA code is needed.

In Sec. 2 we have already mentioned NRQCD [18], which is an EFT of QCD that was developed to exploit the separation of scales

$$mv^2 \ll mv \ll m \quad (10)$$

in a heavy quarkonium. Here, m denotes the heavy quark mass and v stands for the relative velocity of heavy quarks in the quarkonium. The scales m , mv and mv^2 are usually called hard, soft and ultrasoft respectively.

NRQCD is obtained from QCD by integrating out all degrees of freedom above the soft scale. The hard contributions are of course not simply thrown away. Their effects are incorporated in the matching coefficients ω_n that multiply operators \mathcal{O}_n of the NRQCD Lagrangian, which can be schematically written as

$$\mathcal{L}_{\text{NRQCD}} = \sum_n \frac{\omega_n}{m^n} \mathcal{O}_n. \quad (11)$$

Since for charm and bottom quarks we have

$$m \gg \Lambda_{\text{QCD}}, \quad (12)$$

with Λ_{QCD} being the QCD scale at which the perturbation theory breaks down, the matching can be always done perturbatively.

The matching coefficients are fixed by comparing suitable quantities in perturbative QCD and in perturbative NRQCD at finite order in the expansion in v . The NRQCD Lagrangian itself contains an infinite number of operators of arbitrary high dimensions that are compatible with the symmetries of QCD. Using the power counting rules of the theory, one can estimate the relative importance of the operators for each process of interest. For this reason, usually only a small number of NRQCD operators needs to be considered in a practical calculation.

In the following we want to use FEYN CALC to perform the matching between QCD and NRQCD in order to extract the matching coefficients (at leading order in α_s) that enter the decay rate of $\chi_{c0,2} \rightarrow \gamma\gamma$ at leading order

in v . Notice that the decay $\chi_{c_1} \rightarrow \gamma\gamma$ does not occur, because it is forbidden by the Landau-Yang theorem.

These matching coefficients have been calculated in the framework of NRQCD multiple times [18, 22, 52, 53, 54], with many of these calculations carried out in a fully covariant way using the covariant projector technique [23]. Nevertheless, for pedagogical reasons we want to stick to the explicit non-covariant matching in the spirit of [18] and [53]. We also would like to remark that the projector technique has not yet been generalized for higher quarkonium Fock states, that include not only two heavy quarks $|Q\bar{Q}\rangle$ but also gluons (e.g. $|Q\bar{Q}g\rangle$ or $|Q\bar{Q}gg\rangle$). For this reason, the presented approach might still be useful in calculations, where such higher order contributions have to be considered.

The factorization formulas for the decay rates [18] are given by

$$\Gamma(\chi_{c_0} \rightarrow \gamma\gamma) = \frac{2\text{Im}f_{em}(^3P_0)}{3m^4} \langle \chi_{c_0} | \chi^\dagger(-\frac{i}{2}\overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma})\psi|0\rangle \langle 0 | \psi^\dagger(-\frac{i}{2}\overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma})\chi | \chi_{c_0} \rangle \quad (13)$$

$$\Gamma(\chi_{c_2} \rightarrow \gamma\gamma) = \frac{2\text{Im}f_{em}(^3P_2)}{m^4} \langle \chi_{c_2} | \chi^\dagger(-\frac{i}{2}\overleftrightarrow{\mathbf{D}}^{(i}\boldsymbol{\sigma}^j)\psi|0\rangle \langle 0 | \psi^\dagger(-\frac{i}{2}\overleftrightarrow{\mathbf{D}}^{(i}\boldsymbol{\sigma}^j)\chi | \chi_{c_2} \rangle. \quad (14)$$

Here, Pauli spinor field ψ (χ) annihilates (creates) a heavy quark (antiquark), $\boldsymbol{\sigma}$ is the Pauli vector and the covariant derivative is defined as

$$D^\mu = \partial^\mu + igA^\mu \equiv (D^0, -\mathbf{D}), \quad (15)$$

so that

$$iD^0 = i\partial^0 - gA^0, \quad (16)$$

$$i\mathbf{D} = i\nabla + g\mathbf{A}, \quad (17)$$

where A^μ is the gluon field and g stands for the QCD coupling constant. Furthermore,

$$\psi^\dagger \overleftrightarrow{\mathbf{D}} \chi \equiv \psi^\dagger(\mathbf{D}\chi) - (\mathbf{D}\psi)^\dagger\chi, \quad (18)$$

$$\overleftrightarrow{\mathbf{D}}^{(i}\boldsymbol{\sigma}^j) \equiv \frac{1}{2} \left(\overleftrightarrow{\mathbf{D}}^i\boldsymbol{\sigma}^j + \overleftrightarrow{\mathbf{D}}^j\boldsymbol{\sigma}^i \right) - \frac{1}{3}\delta^{ij}\overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma}. \quad (19)$$

The NRQCD long distance matrix elements (LDME)

$$\langle \chi_{c_0} | \chi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma}) \psi | 0 \rangle \langle 0 | \psi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma}) \chi | \chi_{c_0} \rangle \quad (20)$$

and

$$\langle \chi_{c_2} | \chi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}}^{(i} \boldsymbol{\sigma}^{j)}) \psi | 0 \rangle \langle 0 | \psi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}}^{(i} \boldsymbol{\sigma}^{j)}) \chi | \chi_{c_2} \rangle \quad (21)$$

are non-perturbative. They can be determined from fitting to the experimental data or computed on the lattice. On the other hand, the matching coefficients $f_{em}({}^3P_0)$ and $f_{em}({}^3P_2)$ can be calculated in perturbation theory from the matching condition [18]

$$\begin{aligned} & 2 \operatorname{Im} A(Q\bar{Q} \rightarrow Q\bar{Q}) \Big|_{\text{pert. QCD}} \\ &= \frac{2 \operatorname{Im} f_{em}({}^3P_0)}{3m^4} \langle Q\bar{Q} | \chi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma}) \psi | 0 \rangle \langle 0 | \psi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}} \cdot \boldsymbol{\sigma}) \chi | Q\bar{Q} \rangle \Big|_{\text{pert. NRQCD}} \\ &+ \frac{2 \operatorname{Im} f_{em}({}^3P_2)}{m^4} \langle Q\bar{Q} | \chi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}}^{(i} \boldsymbol{\sigma}^{j)}) \psi | 0 \rangle \langle 0 | \psi^\dagger(-\frac{i}{2} \overleftrightarrow{\mathbf{D}}^{(i} \boldsymbol{\sigma}^{j)}) \chi | Q\bar{Q} \rangle \Big|_{\text{pert. NRQCD}}, \end{aligned} \quad (22)$$

where on the right hand side we have displayed only spin triplet terms that contribute at leading order in v . The left hand side of Eq. 22 denotes twice the imaginary part of the perturbative QCD amplitude $Q\bar{Q} \rightarrow Q\bar{Q}$ with 2 photons in the intermediate state. It is understood that this amplitude also has to be expanded to second order in v .

We start the matching calculation by considering the on-shell amplitude for the perturbative process $Q(p_1)\bar{Q}(p_2) \rightarrow \gamma(k_1)\gamma(k_2)$ in QCD. The kinematics of this process reads

$$p_1 + p_2 = k_1 + k_2, \quad (23)$$

$$p_1^2 = p_2^2 = m^2, \quad (24)$$

$$k_1^2 = k_2^2 = 0, \quad (25)$$

with $p_i = (\sqrt{m^2 + \mathbf{p}_i^2}, \mathbf{p}_i) \equiv (E_i, \mathbf{p}_i)$ and $k_i = (|\mathbf{k}_i|, \mathbf{k}_i)$. Obviously, it is most convenient to work in the quarkonium rest frame, where

$$\mathbf{p}_1 = -\mathbf{p}_2 \equiv \mathbf{q}, \quad (26)$$

$$E_1 = E_2 \equiv E_q = \sqrt{m + \mathbf{q}^2}, \quad (27)$$

$$\mathbf{k}_1 = -\mathbf{k}_2. \quad (28)$$

For convenience, the photon polarization vectors can be chosen to be purely spatial, satisfying

$$\epsilon(k_1)^0 = \epsilon(k_2)^0 = 0, \quad (29)$$

$$\epsilon(k_{1/2}) \cdot \mathbf{k}_{1/2} = \epsilon(k_{2/1}) \cdot \mathbf{k}_{2/1} = 0. \quad (30)$$

We need to expand the QCD amplitude in v , i.e. in $|\mathbf{q}|/m$ up to second order which involves rewriting Dirac spinors for Q and \bar{Q} in terms of the Pauli spinors. For the latter let us recall that in 4-dimensions we can decompose any chain of Dirac matrices into scalar, pseudoscalar, vector, axial vector and tensor (SPVAT) components. This decomposition stems from the fact that the 4 dimensional matrices I , γ^5 , γ^μ , $\gamma^5\gamma^\mu$ and $\sigma^{\mu\nu} = \frac{i}{2}[\gamma^\mu, \gamma^\nu]$ form a basis, such that any 4×4 matrix M can be written as

$$M = c_1 I + c_2 \gamma^5 + c_{3\mu} \gamma^\mu + c_{4\mu} \gamma^5 \gamma^\mu + c_{5\mu\nu} \sigma^{\mu\nu}. \quad (31)$$

Therefore, there are only 5 unique spinor structures involving heavy quarks that we can encounter in any tree level amplitude. In fact, the only components that appear in this calculation are vector and axial vector, so that we do not need to consider the other three. Using the explicit form of the Dirac spinors with the non-relativistic normalization,

$$u(\mathbf{q}) = \sqrt{\frac{E_q + m}{2E_q}} \begin{pmatrix} \xi \\ \frac{\mathbf{q} \cdot \boldsymbol{\sigma}}{E_q + m} \xi \end{pmatrix}, \quad (32)$$

$$v(-\mathbf{q}) = \sqrt{\frac{E_q + m}{2E_q}} \begin{pmatrix} -\frac{\mathbf{q} \cdot \boldsymbol{\sigma}}{E_q + m} \eta \\ \eta \end{pmatrix}, \quad (33)$$

with ξ and η being 2-component spinors, we obtain

$$\bar{v}(-\mathbf{q}) \gamma^0 u(\mathbf{q}) = 0, \quad (34)$$

$$\bar{v}(-\mathbf{q}) \gamma^i v(\mathbf{q}) = \eta^\dagger \boldsymbol{\sigma}^i \xi - \frac{\mathbf{q}^i}{2m^2} \eta^\dagger \mathbf{q} \cdot \boldsymbol{\sigma} \xi + \mathcal{O}((|\mathbf{q}|/m)^3), \quad (35)$$

$$\bar{v}(-\mathbf{q}) \gamma^0 \gamma^5 u(\mathbf{q}) = \eta^\dagger \xi \left(1 - \frac{\mathbf{q}^2}{2m^2} \right) + \mathcal{O}((|\mathbf{q}|/m)^3), \quad (36)$$

$$\bar{v}(-\mathbf{q}) \gamma^i \gamma^5 u(\mathbf{q}) = \frac{i}{m} \eta^\dagger (\mathbf{q} \times \boldsymbol{\sigma})^i \xi + \mathcal{O}((|\mathbf{q}|/m)^3). \quad (37)$$

Let us first ignore all the complications related to the non-relativistic expansion and see how far we can get with the QCD amplitude without breaking the covariant notation.

At this order in v and α_s , there are only two tree level diagrams to consider that can be trivially generated with FEYNARTS.

```
In[1]:= $LoadFeynArts = True;
        $FeynCalcStartupMessages = False;
        << FeynCalc';
        $FAVerbose = 0;
```

```
In[2]:= diags = InsertFields[CreateTopologies[0, 2 -> 2],
        {F[3, {2, a}], -F[3, {2, b}]} -> {V[1], V[1]},
        InsertionLevel -> {Classes}, Model -> "SMQCD"];
```

Then the amplitudes are converted into FEYN CALC notation and simplified using standard FEYN CALC functions.

```
In[3]:= amps = (9/4 EQ^2*FCFAConvert[
        CreateFeynAmp[diags, Truncated -> False, PreFactor -> -1],
        IncomingMomenta -> {p1, p2}, OutgoingMomenta -> {k1, k2},
        UndoChiralSplittings -> True,
        TransversePolarizationVectors -> {k1, k2},
        ChangeDimension -> 4, List -> False]) // Contract // Factor
```

```
Out[3]:= iEL^2EQ^2\delta_{ab} \frac{(\varphi(-\bar{p}2,MC)\cdot(\bar{\gamma}\cdot\bar{\epsilon}^*(k1))\cdot(\bar{\gamma}\cdot(\bar{k}1-\bar{p}2)+MC)\cdot(\bar{\gamma}\cdot\bar{\epsilon}^*(k2))\cdot(\varphi(\bar{p}1,MC))}{(\bar{p}2-\bar{k}1)^2-MC^2}
        +iEL^2EQ^2\delta_{ab} \frac{(\varphi(-\bar{p}2,MC)\cdot(\bar{\gamma}\cdot\bar{\epsilon}^*(k2))\cdot(\bar{\gamma}\cdot(\bar{k}2-\bar{p}2)+MC)\cdot(\bar{\gamma}\cdot\bar{\epsilon}^*(k1))\cdot(\varphi(\bar{p}1,MC))}{(\bar{p}2-\bar{k}2)^2-MC^2}
```

The next step is to put the external particles on-shell

```
In[4]:= FCClearScalarProducts[];
        ScalarProduct[k1, k1] = 0;
        ScalarProduct[k2, k2] = 0;
        ScalarProduct[p1, p1] = MC^2;
        ScalarProduct[p2, p2] = MC^2;
```

and perform the SPVAT decomposition of the spinor chains,

```
In[5]:= repRuleHideChains = {
        FCI[Spinor[-p2, MC].GA[x.].GA[5].Spinor[p1, MC]] :> FCI[FV[A, x]],
        FCI[Spinor[-p2, MC].GA[x.].Spinor[p1, MC]] :> FCI[FV[V, x]],
        FCI[Spinor[-p2, MC].GS[x.].GA[5].Spinor[p1, MC]] :> FCI[SP[A, x]],
        FCI[Spinor[-p2, MC].GS[x.].Spinor[p1, MC]] :> FCI[SP[V, x]]
        };
```

```
In[6]:= amps2 = amps // DiracSimplify // DiracReduce // FCI //
```

ReplaceAll[#, repRuleHideChains] & //
PropagatorDenominatorExplicit[#, **Dimension** -> 4] & //
Contract // **ReplaceAll**[#, **Pair**[**Momentum**[k1 | k2],
Momentum[**Polarization**[k1 | k2, ___]] -> 0] &

$$\begin{aligned}
\text{Out}[6] := & \frac{\text{EL}^2 \text{EQ}^2 \delta_{ab} \epsilon^{\overline{Ak1} \epsilon^* (k1) \epsilon^* (k2)}}{2(\overline{k1} \cdot \overline{p2})} - \frac{\text{EL}^2 \text{EQ}^2 \delta_{ab} \epsilon^{\overline{Ak2} \epsilon^* (k1) \epsilon^* (k2)}}{2(\overline{k2} \cdot \overline{p2})} + \frac{i \text{EL}^2 \text{EQ}^2 \delta_{ab} (\overline{V} \cdot \overline{\epsilon}^* (k1)) (\overline{p2} \cdot \overline{\epsilon}^* (k2))}{\overline{k2} \cdot \overline{p2}} \\
& + \frac{i \text{EL}^2 \text{EQ}^2 \delta_{ab} (\overline{p2} \cdot \overline{\epsilon}^* (k1)) (\overline{V} \cdot \overline{\epsilon}^* (k2))}{\overline{k1} \cdot \overline{p2}} + \frac{i \text{EL}^2 \text{EQ}^2 \delta_{ab} (\overline{k1} \cdot \overline{V}) (\overline{\epsilon}^* (k1) \cdot \overline{\epsilon}^* (k2))}{2(\overline{k1} \cdot \overline{p2})} + \frac{i \text{EL}^2 \text{EQ}^2 \delta_{ab} (\overline{k2} \cdot \overline{V}) (\overline{\epsilon}^* (k1) \cdot \overline{\epsilon}^* (k2))}{2(\overline{k2} \cdot \overline{p2})}
\end{aligned}$$

where for convenience we chose to abbreviate vector and axial vector chains as

$$V^\mu \equiv \bar{v}(\mathbf{p}_2) \gamma^\mu u(\mathbf{p}_1), \quad (38)$$

$$A^\mu \equiv \bar{v}(\mathbf{p}_2) \gamma^\mu \gamma^5 u(\mathbf{p}_1). \quad (39)$$

If we are to expand the resulting expression in $|\mathbf{q}|/m$, we must make the \mathbf{q} -dependence explicit in all parts of the amplitude. Since different components of the 4-vectors and spinor chains that appear in the computation depend on $|\mathbf{q}|$ in a different way, it now becomes necessary to break the covariant notation. However, by doing so in a naive way, e.g. by writing something like

$$V \cdot k_1 = V^0 |\mathbf{k}| - \mathbf{V} \cdot \mathbf{k}, \quad (40)$$

$$\begin{aligned}
\epsilon^{\mu\nu\rho\sigma} k_{1\mu} A_\nu \epsilon_\rho^*(k_1) \epsilon_\sigma^*(k_2) &= \epsilon^{\mu 0 \rho \sigma} k_{1\mu} A^0 \epsilon_\rho^*(k_1) \epsilon_\sigma^*(k_2) - \\
&\epsilon^{\mu i \rho \sigma} k_{1\mu} A^i \epsilon_\rho^*(k_1) \epsilon_\sigma^*(k_2)
\end{aligned} \quad (41)$$

we introduce new objects that carry Cartesian indices and thus cannot be handled by the built-in routines for working with Lorentz tensors (e.g. **Contract**, **ScalarProduct**, **ExpandScalarProduct** etc.). Fortunately, it is possible to completely avoid introducing any Cartesian tensors or tensors with mixed Lorentz and Cartesian indices by exploiting FEYNCalc's built-in **TensorFunction** in a clever way.

This approach is based on [55], although we do not consider a boosted $Q\bar{Q}$ -system and assume that the quarkonium is at rest. To see how this works, let us first define a symmetric tensor $E^{\mu\nu}$ with

$$E^{\mu\nu} = \begin{cases} 0 & \text{for } \mu = 0 \text{ or } \nu = 0, \\ \delta^{ij} & \text{for } \mu \neq 0 \text{ and } \nu \neq 0. \end{cases} \quad (42)$$

With $E^{\mu\nu}$ we can write any Cartesian scalar product $x^i y^i$ as

$$x^i y^i = x^i y^j \delta^{ij} = E^{\mu\nu} x_\mu y_\nu \equiv E(x, y), \quad (43)$$

where x^0 and y^0 can be anything, since they drop out by construction. If x is a pure Cartesian vector, then we can choose $x^\mu = (0, x^i)$. Suppose that we want to expand $x^i y^i$ in $|\mathbf{x}|$ or $|\mathbf{y}|$. Then we can write

$$E(x, y) = |\mathbf{x}||\mathbf{y}|E(\hat{x}, \hat{y}), \quad (44)$$

with $x^i = |\mathbf{x}|\hat{x}^i$ and $y^i = |\mathbf{y}|\hat{y}^i$, where $E(\hat{x}, \hat{y})$ does not depend on $|\mathbf{x}|$ and $|\mathbf{y}|$. Therefore, a Minkowski scalar product $x \cdot y$ can be rewritten as

$$x^\mu y_\mu = x^0 y^0 + |\mathbf{x}||\mathbf{y}|E(\hat{x}, \hat{y}) \quad (45)$$

and if x and y are external 4-momenta, then we have

$$x^\mu y_\mu = \sqrt{m_x^2 + |\mathbf{x}|^2} \sqrt{m_y^2 + |\mathbf{y}|^2} + |\mathbf{x}||\mathbf{y}|E(\hat{x}, \hat{y}), \quad (46)$$

so that the expansion in the scalar variables $|\mathbf{x}|$ or $|\mathbf{y}|$ can be carried out without making any reference to 3-vectors. Some useful relations for dealing with E -tensors are

$$E^{\mu\nu} g_{\mu\nu} = -3, \quad (47)$$

$$E^{\mu\nu} E^{\rho\sigma} g_{\mu\rho} = -E^{\nu\sigma} \quad (48)$$

In a similar manner we can also rewrite terms that involve 3-dimensional epsilon tensors by introducing

$$C_{\mu\nu\rho} = \begin{cases} 0 & \text{for } \mu = 0 \text{ or } \nu = 0 \text{ or } \rho = 0, \\ \varepsilon_{ijk} & \text{for } \mu \neq 0 \text{ and } \nu \neq 0 \text{ and } \rho \neq 0, \end{cases} \quad (49)$$

such that

$$\varepsilon^{ijk} x^i y^j z^k = -\varepsilon_{ijk} x^i y^j z^k = -C_{\mu\nu\rho} x^\mu y^\nu z^\rho \equiv -C(x, y, z). \quad (50)$$

Then, it is easy to see that

$$\begin{aligned} \varepsilon^{\sigma\mu\nu\rho} a_\sigma x_\mu y_\nu z_\rho &= \varepsilon^{ijk} (a^0 x_i y_j z_k - x^0 a_i y_j z_k + y^0 a_i x_j z_k - z^0 a_i x_j y_k) \\ &= a^0 C(x, y, z) - x^0 C(a, y, z) + y^0 C(a, x, z) - z^0 C(a, x, y), \end{aligned} \quad (51)$$

from where we again can easily expand in the 3-momenta of a , x , y or z , since

$$C(x, y, z) = |\mathbf{x}||\mathbf{y}||\mathbf{z}|C(\hat{x}, \hat{y}, \hat{z}), \quad (52)$$

with $C(\hat{x}, \hat{y}, \hat{z})$ being independent of $|\mathbf{x}|$, $|\mathbf{y}|$ and $|\mathbf{z}|$. The product of two C tensors can be expressed through

$$C^{\mu\nu\rho}C^{\alpha\beta\gamma} = \begin{vmatrix} E^{\mu\alpha} & E^{\mu\beta} & E^{\mu\gamma} \\ E^{\nu\alpha} & E^{\nu\beta} & E^{\nu\gamma} \\ E^{\rho\alpha} & E^{\rho\beta} & E^{\rho\gamma} \end{vmatrix}. \quad (53)$$

The basic properties of $E^{\mu\nu}$ (denoted as `NRPair`) and $C^{\mu\nu\rho}$ (denoted as `NREps`) can be implemented in `FEYN CALC` with a minimal amount of extra code.

```
In[7]:= SetAttributes[NRPairContract, Orderless];
TensorFunction[NREps, x, y, z];
TensorFunction[{NRPair, "S"}, x, y];
NREps[a___, x_, b___, x_, c___] := 0;
NRPairContract /:
NRPairContract[LorentzIndex[x_], LorentzIndex[x_]] := -3;
NRPairContract /:
NRPairContract[LorentzIndex[x_], y_] *
NRPairContract[LorentzIndex[x_], z_] := - NRPairContract[y, z];
NRPairContract /:
NRPairContract[LorentzIndex[x_], y_] *
NREpsContract[a___, LorentzIndex[x_], b___] := - NREpsContract[a, y, b];
NRPairContract /:
NRPairContract[LorentzIndex[x_], y_]^2 := - NRPairContract[y, y];
NREpsContract /:
NREpsContract[x_, y_, z_]^2 := - 6;
NREpsContract /:
NREpsContract[mu_, nu_, rho_] NREpsContract[al_, be_, ga_] :=
(Det[{{np[mu, al], np[mu, be], np[mu, ga]},
{np[nu, al], np[nu, be], np[nu, ga]},
{np[rho, al], np[rho, be], np[rho, ga]}}] /.
np -> NRPairContract);
```

Contractions of $E^{\mu\nu}$ and $C^{\mu\nu\rho}$ with each other or with the metric tensor are simplified by `NRContract`, while `NRExpand` implements Eq. 51.

```
In[8]:= NRContract[expr_] :=
FixedPoint[(Expand2[Contract[##], {NRPair, NREps}] // {NRPair ->
NRPairContract, NREps -> NREpsContract}) &, expr] /.
{NRPairContract -> NRPair, NREpsContract -> NREps};
```

```
In[9]:= NRExpand[expr_] :=
```

```

FixedPoint[ReplaceRepeated[Expand2[#, {Eps, NREps}],
{Eps[a_Momentum, x_Momentum, y_Momentum, z_Momentum] :=>
  NREN[a] NREps[x, y, z] - NREN[x] NREps[a, y, z] +
  NREN[y] NREps[a, x, z] - NREN[z] NREps[a, x, y]}] &, expr];

```

Here we use NREN to denote the temporal components of 4-momenta. The expansions of spinor chains given in Eqs. 34 - 37 are now straight-forward to translate into FEYN CALC notation.

```

In[10]:= repRuleExpandedChains = {
  Pair[v : Momentum[V], x_] :=> -NRPair[v, x],
  Pair[a : Momentum[A], x_] :=> NREN[a] NREN[x] - NRPair[a, x],
  NREN[Momentum[A]] -> 1 - (qvec^2)/(2 MC^2),
  NREN[Momentum[V]] -> 0,
  NRPair[x_, Momentum[V]] :=> -((qvec^2 NRPair[Momentum[qhat],
    Momentum[{S, I}]] NRPair[Momentum[qhat], x])/(2 MC^2)) +
  NRPair[Momentum[{S, I}], x],
  NRPair[x_, Momentum[A]] :=>
  -((I qvec NREps[Momentum[qhat], Momentum[{S, I}], x])/MC),
  NREps[x_, a : Momentum[A], y_] :=> (li =
  LorentzIndex[Unique[]]; -NREps[x, li, y] NRPair[a, li]),
  NREps[x_, v : Momentum[V], y_] :=>
  (li = LorentzIndex[Unique[]]; -NREps[x, li, y] NRPair[v, li])
};

```

Simplifications that are specific to the kinematics of the process are also easy to define.

```

In[11]:= NREN[Momentum[Polarization[k1 | k2, ___]]] = 0;
  NREN[Momentum[k1 | k2]] = kvec;
  NREN[Momentum[p1 | p2]] = Sqrt[MC^2 + qvec^2];
  NREN[Momentum[qhat | qhatp | {S, -}]] = 0;
  NREN[Momentum[k1hat]] = 1;
  NRPair[Momentum[p1], x_] = qvec NRPair[Momentum[qhat], x];
  NRPair[Momentum[p2], x_] = -qvec NRPair[Momentum[qhat], x];
  NRPair[Momentum[p1p], x_] = qvec NRPair[Momentum[qhatp], x];
  NRPair[Momentum[p2p], x_] = -qvec NRPair[Momentum[qhatp], x];
  NRPair[Momentum[k1 | k2 | k1hat | k2hat],
    Momentum[Polarization[k2 | k1, ___]]] = 0;
  NRPair[Momentum[x_], Momentum[x_]] :=
  1 /; MemberQ[{qhat, qhatp, p1hat, p2hat, k1hat, k2hat}, x];
  NRPair[Momentum[k1], x_] = kvec NRPair[Momentum[k1hat], x];
  NRPair[Momentum[k2], x_] = -kvec NRPair[Momentum[k1hat], x];
  kvec = Sqrt[MC^2 + qvec^2];

  repRuleExpansion = {
    FCI@SP[x_, a : Polarization[z_, ___]] /; MemberQ[{k1, k2}, z] :=>

```

```

–NRPair[Momentum[x], Momentum[a]],
FCI@SP[x_, (y : p1 | p2 | k1 | k2 | {S, I} | {S, -I} | k1hat | qhat)] :=
NREn[Momentum[x]] NREn[Momentum[y]] –
NRPair[Momentum[x], Momentum[y]],
NREps[a___, Momentum[k1], z___] := kvec NREps[a, Momentum[k1hat], z],
NREps[a___, Momentum[k2], z___] := – kvec NREps[a, Momentum[k1hat],
z],
NREps[a___, Momentum[p1], b___] := qvec NREps[a, Momentum[qhat], b],
NREps[a___, Momentum[p2], b___] := –qvec NREps[a, Momentum[qhat], b
]
};

```

Finally, we can expand the amplitude up to second order in $|\mathbf{q}|/m$.

```

In[12]:= amps3 = amps2 // NRExpand //
ReplaceRepeated[#, repRuleExpandedChains] & // NRContract //
ReplaceRepeated[#, repRuleExpansion] & // Series[#, {qvec, 0, 2}] & //
Normal // PowerExpand // NRContract;

```

To obtain $2 \text{Im} A(Q(p'_1)\bar{Q}(p'_2) \rightarrow Q(p_1)\bar{Q}(p_2))$ from our expanded amplitude, we need to multiply $A(Q(p'_1)\bar{Q}(p'_2) \rightarrow \gamma(k_1)\gamma(k_2))$ by $A^*(Q(p_1)\bar{Q}(p_2) \rightarrow \gamma(k_1)\gamma(k_2))$, sum over polarizations of the external photons and perform the phase space integration. For the latter we can use that

$$\int d\Omega_{k_1} \hat{k}_1^{i_1} \dots \hat{k}_1^{i_{2n+1}} = 0, \quad (54)$$

$$\int d\Omega_{k_1} \hat{k}_1^{i_1} \dots \hat{k}_1^{i_{2n}} = \frac{4\pi}{(n+2)!!} (\delta^{i_1 i_2} \dots \delta^{i_{2n-1} i_{2n}} + \text{permutations}). \quad (55)$$

To implement these relations we need an auxiliary function that uncontracts the indices of \hat{k}_1

```

In[13]:= NRUncontract[expr_, l_List] :=
expr /. {Power[t_NRPair, n_] := times @@ Table[t, {i, 1, n}],
Power[t_NREps, n_] := times @@ Table[t, {i, 1, n}} // . {
NRPair[y_, x_] /; ! FreeQ2[y, 1] && Head[x] != LorentzIndex :=>
(li = Unique[$AL]; –NRPair[y, LorentzIndex[li]] NRPair[x, LorentzIndex[li]]),
NREps[w___, y_, x___] /; ! FreeQ2[y, 1] :=> (li = Unique[$AL];
–NRPair[y, LorentzIndex[li]] NREps[w, LorentzIndex[li], x])
} /. times -> Times;

```

and a replacement rule that handles the angular integration

```

In[14]:= angularIntegration[hat_] := {
qHead[NRPair[i_, Momentum[hat]]] /; FreeQ2[{i}, {hat, S}] :=> 0,
qHead[a_Times] :=> qHead[(List @@ a) /. NRPair[Momentum[hat], b_] :=>

```

```

{hat, b /. LorentzIndex[c_, -] := c}],
qHead[a_List] := (Tdec[a, {}], List -> False, FCE -> False,
  Dimension -> 3] //. {(h : LorentzIndex | Momentum)[x_, 3] :=
  h[x], Pair -> NRPair}
};

```

Then the left hand side of Eq. 22 is given by

```

In[15]:= res = (1/(16 Pi)) (Collect[(amps3 /. qhat -> qhatp)*
  ComplexConjugate[amps3] /. NRPair[x_, y_] :=
  -FCI@SP[x, y] + NREN[x] NREN[y], qvec] /. qvec^4 -> 0) //
  DoPolarizationSums[#, k1, k2] & //
  DoPolarizationSums[#, k2, k1] & //
  ReplaceRepeated[#, repRuleExpansion] & // Cancel //
  NRUncontract[#, {k1hat}] & //
  FCLoopIsolate[#, {k1hat}, Head -> qHead] & //
  ReplaceRepeated[#, angularIntegration[k1hat]] & // NRContract //
  ReplaceAll[#, {EL^4 -> 16 Pi^2 AlphaFS^2}] & //
  SelectNotFree[#, S] &

```

```

Out[15]:= 
$$\frac{4\pi\alpha^2EQ^4qvec^2\delta_{ab}NRPair(\overline{qhat,\{S,i\}})NRPair(\overline{qhatp,\{S,-i\}})}{5MC^4} +$$


$$\frac{22\pi\alpha^2EQ^4qvec^2\delta_{ab}NRPair(\overline{qhat,\{S,-i\}})NRPair(\overline{qhatp,\{S,i\}})}{15MC^4} +$$


$$\frac{4\pi\alpha^2EQ^4qvec^2\delta_{ab}NRPair(\overline{qhat,qhatp})NRPair(\overline{\{S,-i\},\{S,i\}})}{5MC^4}$$


```

An explicit expression for the right hand side of Eq. 22 can be obtained by using Fourier decompositions of the Pauli spinor fields (c.f. [56]), so that we end up with

$$\begin{aligned}
& \frac{4\alpha^2Q^4\pi}{5m^4} \mathbf{q} \cdot \mathbf{q}' \eta^\dagger \boldsymbol{\sigma} \xi \xi^\dagger \boldsymbol{\sigma} \eta + \frac{4\alpha^2Q^4\pi}{5m^4} \eta^\dagger \mathbf{q} \cdot \boldsymbol{\sigma} \xi \xi^\dagger \mathbf{q}' \cdot \boldsymbol{\sigma} \eta + \frac{22\alpha^2Q^4\pi}{15m^4} \eta^\dagger \mathbf{q}' \cdot \boldsymbol{\sigma} \xi \xi^\dagger \mathbf{q} \cdot \boldsymbol{\sigma} \eta \\
& = \frac{\text{Im}f_{em}(^3P_2)}{m^4} \mathbf{q} \cdot \mathbf{q}' \eta^\dagger \boldsymbol{\sigma} \xi \xi^\dagger \boldsymbol{\sigma} \eta + \frac{\text{Im}f_{em}(^3P_2)}{m^4} \eta^\dagger \mathbf{q} \cdot \boldsymbol{\sigma} \xi \xi^\dagger \mathbf{q}' \cdot \boldsymbol{\sigma} \eta \\
& + \frac{2}{3} \frac{(\text{Im}f_{em}(^3P_0) - \text{Im}f_{em}(^3P_2))}{m^4} \eta^\dagger \mathbf{q}' \cdot \boldsymbol{\sigma} \xi \xi^\dagger \mathbf{q} \cdot \boldsymbol{\sigma} \eta, \tag{56}
\end{aligned}$$

from which we can immediately read off the values of the matching coefficients

$$\text{Im}f_{em}(^3P_0) = 3\alpha^2Q^4\pi, \tag{57}$$

$$\text{Im}f_{em}(^3P_2) = \frac{4}{5}\alpha^2Q^4\pi, \tag{58}$$

that agree with the known results from the literature [18, 22, 52, 53, 54].

5. Summary

We have presented new features and improvements in FEYNALC 9.0 and discussed cases in which FEYNALC can be used to obtain new results. Although the very first version of FEYNALC appeared almost 25 years ago, the development is still far from being complete. New developments in theoretical particle physics show possible directions in which FEYNALC can evolve. This includes better support for multi-loop calculations and determination of matching coefficients in effective field theories, but also built-in interfaces to other useful software tools and the ability to work with non-relativistic theories.

Finally, we would like to observe that in the last two years some new general-purpose packages [57, 58] for QFT calculations were released, which follow the approach similar to that of FEYNALC and thus provide a comparable level of flexibility. This development shows that even in the age of fully automatic packages for 1-loop calculations, user-friendly, semi-automatic tools like FEYNALC are still in demand and employed in many interesting research projects.

Acknowledgments

Two of the authors (RM and FO) would like to thank Daniel Wyler for his help and support in the earlier days of FEYNALC.

One of the authors (VS) wants to thank to Hector Martinez Neira for bringing his attention to FEYNALC for the first time and to his PhD supervisor Nora Brambilla for encouraging him to work in this direction. Simone Biondini is acknowledged for testing 1-loop tensor decompositions with TID. VS would also like to express his gratitude to Antonio Vairo, Christoph Both, Thomas Hahn, Georg Weiglein, Sergey Larin, Claude Duhr, Matthias Steinhauser, Alexander Smirnov, Massimo Passera and Yu Jia for useful discussions and explanations. His work has been supported by the DFG and the NSFC through funds provided to the Sino-German CRC 110 “Symmetries and the Emergence of Structure in QCD”, and by the DFG cluster of excellence “Origin and structure of the universe” (www.universe-cluster.de).

Last but not least, all the authors would like to thank to the participants of the FEYNALC mailing list for their bug reports, feature requests, suggestions and encouragements.

References

- [1] C. Anastasiou, C. Duhr, F. Dulat, F. Herzog, and B. Mistlberger, Higgs boson gluon-fusion production in N³LO QCD, *Phys. Rev. Lett.*, **114**, 212001, (2015), [arXiv:1503.06056](#).
- [2] P. Marquard, A. V. Smirnov, V. A. Smirnov, and M. Steinhauser, Quark mass relations to four-loop order, *Phys. Rev. Lett.*, **114**, 142002, (2015), [arXiv:1502.01030](#).
- [3] M. Beneke, Y. Kiyo, P. Marquard, A. Penin, J. Piclum, D. Seidel, and M. Steinhauser, Leptonic decay of the Upsilon(1S) meson at third order in QCD, *Phys. Rev. Lett.*, **112**, 151801, (2014), [arXiv:1401.3005](#).
- [4] R. Mertig, M. Böhm, and A. Denner, Feyn Calc - Computer-algebraic calculation of Feynman amplitudes, *Comput. Phys. Commun.*, **64**, 345–359, (1991).
- [5] A. Denner, J. Küblbeck, R. Mertig, and M. Böhm, Electroweak radiative corrections to $e^+e^- \rightarrow HZ$, *Z. Phys. C*, **56**, 261–272, (1992).
- [6] W. Beenakker, A. Denner, S. Dittmaier, R. Mertig, and T. Sack, High-energy approximation for on-shell W-pair production, *Nucl. Phys. B*, **410**, 245–279, (1993).
- [7] W. Beenakker, A. Denner, W. Hollik, R. Mertig, T. Sack, and D. Wackerroth, Electroweak one-loop contributions to top pair production in hadron colliders, *Nucl. Phys. B*, **411**, 343–380, (1994).
- [8] W. Beenakker, A. Denner, S. Dittmaier, and R. Mertig, On-shell W-pair production in the TeV range, *Phys. Lett. B*, **317**, 622–630, (1993).
- [9] R. Mertig and W. L. van Neerven, The calculation of the two-loop spin splitting functions $P_{ij}^{(1)}(x)$, *Z. Phys. C*, **70**, 637–653, (1996), [arXiv:hep-ph/9506451](#).
- [10] M. Buechler, G. Colangelo, J. Kambor, and F. Orellana, Dispersion relations and soft pion theorems for $K \rightarrow \pi\pi$, *Phys. Lett. B*, **521**, 22–28, (2001), [arXiv:hep-ph/0102287](#).

- [11] T. Hahn, Generating Feynman Diagrams and Amplitudes with FeynArts 3, *Comput. Phys. Commun.*, **140**, 418–431, (2001), [arXiv:hep-ph/0012260](#).
- [12] F. Feng and R. Mertig, FormLink/FeynCalcFormLink : Embedding FORM in Mathematica and FeynCalc, (2012), [arXiv:1212.3522](#).
- [13] T. Hahn and M. Perez-Victoria, Automatized One-Loop Calculations in 4 and D dimensions, *Comput. Phys. Commun.*, **118**, 153–165, (1999), [arXiv:hep-ph/9807565](#).
- [14] G. Cullen, H. van Deurzen, N. Greiner, G. Heinrich, G. Luisoni, P. Mastrolia, E. Mirabella, G. Ossola, T. Peraro, J. Schlenk, J. F. von Soden-Fraunhofen, and F. Tramontano, GoSam-2.0: a tool for automated one-loop calculations within the Standard Model and beyond, *Eur. Phys. J. C*, **74**, 8, 3001, (2014), [arXiv:1404.7096](#).
- [15] G. Belanger, F. Boudjema, J. Fujimoto, T. Ishikawa, T. Kaneko, K. Kato, and Y. Shimizu, GRACE at ONE-LOOP: Automatic calculation of 1-loop diagrams in the electroweak theory with gauge parameter independence checks, *Phys. Rept.*, **430**, 117–209, (2006), [arXiv:hep-ph/0308080](#).
- [16] M. Tentyukov and J. Fleischer, A Feynman Diagram Analyser DIANA, *Comput. Phys. Commun.*, **132**, 124–141, (2000), [arXiv:hep-ph/9904258](#).
- [17] J. A. M. Vermaseren, New features of FORM, (2007), [arXiv:math-ph/0010025](#).
- [18] G. T. Bodwin, E. Braaten, and G. P. Lepage, Rigorous QCD Analysis of Inclusive Annihilation and Production of Heavy Quarkonium, *Phys. Rev. D*, **51**, 1125–1171, (1995), [arXiv:hep-ph/9407339](#).
- [19] N. Brambilla, A. Pineda, J. Soto, and A. Vairo, Potential NRQCD: an effective theory for heavy quarkonium, *Nucl. Phys. B*, **566**, 275, (2000), [arXiv:hep-ph/9907240](#).
- [20] P. Cho and A. Leibovich, Color-octet quarkonia production, *Phys. Rev. D*, **53**, 150–162, (1996), [arXiv:hep-ph/9505329](#).

- [21] Y.-J. Zhang and K.-T. Chao, Double charm production $e^+e^- \rightarrow J/\psi + c\bar{c}$ at B factories with next-to-leading order QCD corrections, *Phys. Rev. Lett.*, **98**, 092003, (2007), [arXiv:hep-ph/0611086](#).
- [22] A. Petrelli, M. Cacciari, M. Greco, F. Maltoni, and M. L. Mangano, NLO Production and Decay of Quarkonium, *Nucl. Phys. B*, **514**, 245–309, (1998), [arXiv:hep-ph/9707223](#).
- [23] G. T. Bodwin and A. Petrelli, Order- v^4 Corrections to S-wave Quarkonium Decay, *Phys. Rev. D*, **66**, 094011, (2002), [arXiv:hep-ph/0205210](#).
- [24] Y. Jia, X.-T. Yang, W.-L. Sang, and J. Xu, $\mathcal{O}(\alpha_s v^2)$ correction to pseudoscalar quarkonium decay to two photons, *J. High Energ. Phys.*, **2011**, (2011), [arXiv:1104.1418](#).
- [25] C. O. Dib, R. Rosenfeld, and A. Zerwekh, Double Higgs Production and Quadratic Divergence Cancellation in Little Higgs Models with T Parity, *JHEP*, **0605**, 074, (2006), [arXiv:hep-ph/0509179](#).
- [26] D. Buttazzo, G. Degrassi, P. P. Giardino, G. F. Giudice, F. Sala, A. Salvio, and A. Strumia, Investigating the near-criticality of the Higgs boson, *J. High Energ. Phys.*, **2013**, (2013), [arXiv:1307.3536](#).
- [27] D. de Florian and J. Mazzitelli, Higgs Boson Pair Production at Next-to-Next-to-Leading Order in QCD, *Phys. Rev. Lett.*, **111**, 201801, (2013), [arXiv:1309.6594](#).
- [28] B. Xiao, Y.-K. Wang, Z.-Q. Zhou, and S. hua Zhu, Edge Charge Asymmetry in Top Pair Production at the LHC, *Phys. Rev. D*, **83**, 057503, (2011), [arXiv:1101.2507](#).
- [29] L. S. Geng, J. M. Camalich, L. Alvarez-Ruso, and M. J. V. Vacas, Nucleon-to-Delta axial transition form factors in relativistic baryon chiral perturbation theory, *Phys. Rev. D*, **78**, 014011, (2008), [arXiv:0801.4495](#).
- [30] D. R. Phillips, M. R. Schindler, and R. P. Springer, An effective-field-theory analysis of low-energy parity-violation in nucleon-nucleon scattering, *Nucl. Phys. A*, **822**, 1–19, (2009), [arXiv:0812.2073](#).

- [31] J. Kopp, V. Niro, T. Schwetz, and J. Zupan, DAMA/LIBRA data and leptonically interacting dark matter, *Phys. Rev. D*, **80**, (2009), [arXiv:0907.3159](#).
- [32] J. M. Cline, A. R. Frey, and F. Chen, Metastable dark matter mechanisms for INTEGRAL 511 keV γ rays and DAMA/CoGeNT events, *Phys. Rev. D*, **83**, 083511, (2010), [arXiv:1008.1784](#).
- [33] S. Bray, J. S. Lee, and A. Pilaftsis, Heavy Majorana Neutrino Production at e- gamma Colliders, *Phys. Lett. B*, **628**, 250–261, (2005), [arXiv:hep-ph/0508077](#).
- [34] R. Laha, B. Dasgupta, and J. F. Beacom, Constraints on New Neutrino Interactions via Light Abelian Vector Bosons, *Phys. Rev. D*, **89**, 093025, (2013), [arXiv:1304.3460](#).
- [35] B. Dasgupta and J. Kopp, A ménage à trois of eV-scale sterile neutrinos, cosmology, and structure formation, *Phys. Rev. Lett.*, **112**, 031803, (2013), [arXiv:1310.6337](#).
- [36] S. Foffa and R. Sturani, Dynamics of the gravitational two-body problem at fourth post-Newtonian order and at quadratic order in the Newton constant, *Phys. Rev. D*, **87**, (2013), [arXiv:1206.7087](#).
- [37] J. R. Gaunt and M. Stahlhofen, The Fully-Differential Quark Beam Function at NNLO, *JHEP*, **1412**, 146, (2014), [arXiv:1409.8281](#).
- [38] F. Feng, Y. Jia, and W.-L. Sang, Can NRQCD explain the $\gamma\gamma^* \rightarrow \eta_c$ transition form factor data?, (2015), [arXiv:1505.02665](#).
- [39] A. V. Smirnov and V. A. Smirnov, FIRE4, LiteRed and accompanying tools to solve integration by parts relations, *Comput. Phys. Commun.*, **184**, 2820–2827, (2013), [arXiv:1302.5885](#).
- [40] A. V. Smirnov, FIESTA 3: cluster-parallelizable multiloop numerical calculations in physical regions, *Comput. Phys. Commun.*, **185**, 2090–2100, (2013), [arXiv:1312.3186](#).
- [41] S. Borowka, J. Carter, and G. Heinrich, Numerical evaluation of multi-loop integrals for arbitrary kinematics with SecDec 2.0, *Comput. Phys. Commun.*, **184**, 396–408, (2012), [arXiv:1204.4152](#).

- [42] G. Passarino and M. Veltman, One Loop Corrections for $e^+ e^-$ Annihilation Into $\mu^+ \mu^-$ in the Weinberg Model, Nucl.Phys., **B160**, 151, (1979).
- [43] F. Feng, \$Apart: A Generalized Mathematica Apart Function, Comput. Phys. Commun., **183**, 2158–2164, (2012), [arXiv:1204.2314](#).
- [44] K. Chetyrkin and F. Tkachov, Integration by parts: The algorithm to calculate ϵ -functions in 4 loops, Nucl. Phys. B, **192**, 159–204, (1981).
- [45] R. Mertig and R. Scharf, TARCER - A Mathematica program for the reduction of two-loop propagator integrals, Comput. Phys. Commun., **111**, 265–273, (1998), [arXiv:hep-ph/9801383](#).
- [46] R. N. Lee, Presenting LiteRed: a tool for the Loop InTEgrals REDuction, (2012), [arXiv:1212.2685](#).
- [47] C. Studerus, Reduze - Feynman Integral Reduction in C++, Comput. Phys. Commun., **181**, 1293–1300, (2009), [arXiv:0912.2546](#).
- [48] C. Anastasiou and A. Lazopoulos, Automatic Integral Reduction for Higher Order Perturbative Calculations, JHEP, **0407**, 046, (2004), [arXiv:hep-ph/0404258](#).
- [49] F. Feng, Automated One-loop Computation in Quarkonium Process within NRQCD Framework, (2013), [arXiv:1307.5587](#).
- [50] M. R. Fiorentin, FaRe: a Mathematica package for tensor reduction of Feynman integrals, International Journal of Modern Physics C, p. 1650027, (2015), [arXiv:1507.03527](#).
- [51] J. Küblbeck, M. Böhm, and A. Denner, Feyn arts computer-algebraic generation of Feynman graphs and amplitudes, Comput. Phys. Commun., **60**, 165–180, (1990).
- [52] J. P. Ma and Q. Wang, Corrections For Two Photon Decays of χ_{c0} and χ_{c2} and Color Octet Contributions, Phys. Lett. B, **537**, 233–240, (2002), [arXiv:hep-ph/0203082](#).
- [53] N. Brambilla, E. Mereghetti, and A. Vairo, Electromagnetic quarkonium decays at order v^7 , JHEP, **0608**, 039, (2006), [arXiv:hep-ph/0604190](#).

- [54] W.-L. Sang, F. Feng, Y. Jia, and S.-R. Liang, Next-to-next-to-leading-order QCD corrections to $\chi_{c0,2} \rightarrow \gamma\gamma$, (2015), [arXiv:1511.06288](#).
- [55] E. Braaten and Y.-Q. Chen, Helicity Decomposition for Inclusive J/ψ Production, *Phys. Rev. D*, **54**, 3216–3227, (1996), [arXiv:hep-ph/9604237](#).
- [56] P. Cho and A. K. Leibovich, Color-octet quarkonia production II, *Phys.Rev. D*, **53**, 6203–6217, (1996), [arXiv:hep-ph/9511315](#).
- [57] M. Wiebusch, HEPMath: A Mathematica Package for Semi-Automatic Computations in High Energy Physics, *Comput. Phys. Commun.*, **195**, 172–190, (2014), [arXiv:1412.6102](#).
- [58] H. H. Patel, Package-X: A Mathematica package for the analytic calculation of one-loop integrals, *Comput. Phys. Commun.*, **197**, 276–290, (2015), [arXiv:1503.01469](#).